UFR de Mathématiques et Informatique
45, rue des Saints-Pères 75270 Paris cedex 06

Corrigé de l'examen du 19 mai 2009
L2 : Environnement de calcul scientifique et modélisation

Questions de cours :

1. Grâce à la décomposition le système $Ax = b$ s'écrit $(L * U)x = b$; en posant $y = Lx$, on est d'abord amené à résoudre un système triangulaire inférieur d'inconnue y : $Ly = b$; ensuite on doit résoudre le système triangulaire supérieur $Ux = y$ afin d'obtenir x.

2. La fonction f n'est pas définie en $\{-1, +1\}$. Il faut tenir compte de ces singularités lors de la discrétisation de l'intervalle $[-2, 2]$. Comme f tend vers l'infini au voisinage de -1 et 1, il faut borner le graphe : soit en utilisant `rect`, soit en ne prenant pas de valeurs proches de -1 et 1.
Finalement, on doit utiliser un nombre suffisant de points discrets sur $[-2, 2] \setminus \{-1, +1\}$ afin d'obtenir un graphe lisse.

Exercice 1

- Sans boucle : `A = matrix( [1:n^2], n, n)' ; A(2:2:n,:) = A(2:2:n, n:-1:1)`
- Des solutions avec boucles sont possibles mais plus compliquées et longues.

Exercice 2

On obtient la matrice A par exemple comme suit : `A=matrix( [10:25],4,4)'`
et on rappelle que `v=[2,4]`.

(1) `A(v,v)` extrait les éléments de A qui se trouvent en ligne 2 ou 4 et en colonne 2 ou 4, d'où l'affichage `ans`$= \begin{pmatrix} 15 & 17 \\ 23 & 25 \end{pmatrix}$.

(2) On a `v-1=[1,3]`, `A(v,v-1)` extrait les éléments de A qui se trouvent en ligne 2 ou 4 et en colonne 1 ou 3, d'où `ans`$= \begin{pmatrix} 14 & 16 \\ 22 & 24 \end{pmatrix}$.

(3) Comme `v($:-1:1)=[4,2]`, on extrait les mêmes éléments que dans (1) mais en ordre inverse, `A(v($:-1:1),v($:-1:1))` affiche `ans`$= \begin{pmatrix} 25 & 23 \\ 17 & 15 \end{pmatrix}$.

(4) La commande `A(v,:)=[]` va remplacer toutes les colonnes des lignes données par v, *i.e.* 2 et 4, par la matrice vide, *càd* on enlève la deuxième et la quatrième ligne de A, d'où l'affichage $A = \begin{pmatrix} 10 & 11 & 12 & 13 \\ 18 & 19 & 20 & 21 \end{pmatrix}$

(5) De façon similaire `A(v,v(2))=v'` remplace les éléments des lignes 2 et 4 et colonne `v(2)=4`, par les éléments du vecteur colonne `v'=[ 2 ; 4 ]`, d'où $A = \begin{pmatrix} 10 & 11 & 12 & 13 \\ 14 & 15 & 16 & 2 \\ 18 & 19 & 20 & 21 \\ 22 & 23 & 24 & 4 \end{pmatrix}$.

(6) `A(v,v)=v'*v` remplace les quatre éléments, affichés en (1), par les valeurs de la matrice `v' * v`$= \begin{pmatrix} 2 \\ 4 \end{pmatrix} \begin{pmatrix} 2 & 4 \end{pmatrix} = \begin{pmatrix} 4 & 8 \\ 8 & 16 \end{pmatrix}$. On obtient $A = \begin{pmatrix} 10 & 11 & 12 & 13 \\ 14 & 4 & 16 & 8 \\ 18 & 19 & 20 & 21 \\ 22 & 8 & 24 & 16 \end{pmatrix}$.

Exercice 3

(1) La commande `plot2d` affiche les points de coordonnées $(x(i), y(i))$, pour $1 \leq i \leq 4$, reliés par des traits : ceci donne un triangle de coins $(1,1)$, $(2,1)$ et $(1,2)$.

(2) L'intervalle $[-\pi, +\pi]$ est partagé en 50 points (49 intervalles de même longueur) qui sont placés dans le vecteur colonne `x`. À la fin de la boucle `for`, on a construit une matrice `y` de taille $[50, 3]$ avec $y(i,1) = -\sin(x(i))$, $y(i,2) = 0$ et $y(i,3) = \sin(x(i))$, pour $1 \leq i \leq 50$. On aura donc l'affichage sur l'intervalle $[-\pi, +\pi]$ des fonctions $x \mapsto -\sin(x)$, $x \mapsto 0$ et $x \mapsto \sin(x)$ avec trois couleurs différentes.

(3) Le polynômes `p` est défini par ses coefficients : $p = -1 + x^2$, donc $2p = 2(x^2 - 1)$ dont les racines -1 et $+1$ seront affichées.

(4) `v=(-1)^[0:2]=[1, -1, 1]`, et `[v;v;v]'=`$\begin{pmatrix} 1 & -1 & 1 \\ 1 & -1 & 1 \\ 1 & -1 & 1 \end{pmatrix}' = \begin{pmatrix} 1 & 1 & 1 \\ -1 & -1 & -1 \\ 1 & 1 & 1 \end{pmatrix}$.

Donc `A = diag([10^2,10^3,10^4]) * [v;v;v]'`

$$= \begin{pmatrix} 100 & 0 & 0 \\ 0 & 1000 & 0 \\ 0 & 0 & 10000 \end{pmatrix} \begin{pmatrix} 1 & 1 & 1 \\ -1 & -1 & -1 \\ 1 & 1 & 1 \end{pmatrix} = \begin{pmatrix} 100 & 100 & 100 \\ -1000 & -1000 & -1000 \\ 10000 & 1000 & 10000 \end{pmatrix}.$$

(5) Si l'on pose `X=[x,x]` et `Y=[y,-y]`, alors `X` et `Y` sont des matrices chacune à trois lignes et deux colonnes ; la commande `plot2d([x,x],[y,-y],style=[1,1])` affiche deux courbes avec la même couleur. La première courbe est définie par les points $(-1,0)$, $(0,1)$ et $(1,0)$, tandis que la deuxième est donnée par $(-1,0)$, $(0,-1)$ et $(1,0)$. On obtient ainsi un carré...

Exercice 4

1. On a `det(A)=det(L*D*L')=det(L)det(D)det(L')=det(D)`$= d_1 \ldots d_n$.
Grâce à la décomposition, le calcul du déterminant revient à faire le produit des éléments de D.

2. Grâce à la décomposition, le système linéaire s'écrit $(L * D * L') * x = b$. En un premier temps on va résoudre le système triangulaire inférieur $L * y = b$, ensuite le système triangulaire supérieur, $L' * x = \tilde{y}$, où $\tilde{y} = D^{-1} y = [y_1/d_{11}, \ldots, y_n/d_{nn}]'$.

3. On propose une fonction `decomp` avec tests qui profite des opérations par blocs afin d'éliminer des boucles. Elle s'inspire de la fonction `mon_lu()` traitée en TP.

```
function [L,D]=decomp(A)
 n=size(A,1)
 if (n<>size(A,2)) then
    error("Matrice non carree");
 end
 if  (max(abs(A-A'))<>0.0) then
  error("Matrice non symetrique");
 end

 L=eye(A);
 d=zeros(n,1);
 v=zeros(n,1)
 for j=1:n
  v(1:j-1)= L(j,1:j-1)' .* d(1:j-1)
  v(j)= A(j,j) - L(j, 1:j-1) * v(1:j-1)

  d(j)=v(j)
  if (d(j)==0.0) then
    error("Element diagonal nul !")
  end
  L(j+1:n,j) = ( A(j+1:n,j) - L(j+1:n, 1:j-1) * v(1:j-1) ) / v(j)
 end
 D=diag(d)
endfunction
```